
Function f(X As Double) As Double
   f = 2 * Sin(2 * X) + 3 * Cos(X) - X

End Function

Function df(X As Double) As Double
   df = 4 * Cos(2 * X) - 3 * Sin(X) - 1

End Function

Function fText() As String
   fText = "2Sin(2X)+3Cos(X)-X"

End Function

Option Explicit

Function Bisect(a As Double, b As Double, eps As Double, _
               itmax As Integer, ierr As Integer) As Double

' Menghitung akar dengan metoda bisection
' Input: a, b adalah dua bilangan yang mengapit akar
'        Eps adalah ketelitian yang dikehendaki
' Output:
'        Bisect adalah hasil hitungan akar dengan ketelitian Eps
'        ierr adalah kode error
'        ierr = 1, warning, akar belum memenuhi ketelitian,
'                  namun maksimum iterasi sudah tercapai
'        ierr = 0, sukses mencari akar

Dim c As Double, NoIter As Integer

' Sediakan variabel untuk counter
   NoIter = 1

Start:
'   Hitung nilai tengah
   c = (a + b) / 2
   

'   Cek apakah akar sudah memenuhi syarat,
'   jika memenuhi syarat: catat akarnya dan keluar
   Bisect = c
   If Abs(b - c) <= eps Then
       ierr = 0
       Exit Function
   End If

' Jika iterasi terlampaui, maka selesailah tugas kita
   If NoIter = itmax Then
       ierr = 1
       Exit Function
   End If
   

'   Naikkan counter
   NoIter = NoIter + 1
   

'   Tentukan apakah c menggantikan a atau b
   If Sgn(f(b)) * Sgn(f(c)) <= 0 Then
      a = c
   Else
      b = c
   End If

'   Kembali ke atas
   GoTo Start

End Function

Function Newton(X0 As Double, eps As Double, itmax As Integer, ierr As Integer) As Double

' Menghitung akar dengan metoda Newton
' Input: X0 adalah tebakan awal
'        Eps adalah ketelitian yang dikehendaki
'        Itmax adalah iterasi maksimum yang akan dilakukan
' Output:
'        Newton adalah hasil hitungan akar dengan ketelitian Eps
'        ierr adalah kode error
'        ierr = 2, fatal error, penyebut = 0

'        ierr = 1, warning, akar belum memenuhi ketelitian,
'                  namun maksimum iterasi sudah tercapai
'        ierr = 0, sukses mencari akar

Dim NoIter As Integer, Penyebut As Double, X1 As Double

' Sediakan variabel untuk counter
   NoIter = 1
   

Langkah3:
' Hitung penyebut pada metoda Newton
   Penyebut = df(X0)

' Jika penyebut = 0, maka error terjadi
   If Penyebut = 0# Then
       ierr = 2
       Exit Function
   End If
   

' Mulai metoda Newton
   X1 = X0 - f(X0) / Penyebut
   Newton = X1
   

' Test konvergensi
   If Abs(X1 - X0) <= eps Then
       ierr = 0
       Exit Function
   End If

' Jika iterasi terlampaui, maka selesailah tugas kita
   If NoIter = itmax Then
       ierr = 1
       Exit Function
   End If
  

' Jika belum memenuhi kriteria lakukan hitungan lagi
   NoIter = NoIter + 1
   X0 = X1
   GoTo Langkah3
   

End Function

Function Secant(X0 As Double, X1 As Double, eps As Double, _
               itmax As Integer, ierr As Integer) As Double

' Menghitung akar dengan metoda Newton
' Input: X0 adalah tebakan awal
'        Eps adalah ketelitian yang dikehendaki
'        Itmax adalah iterasi maksimum yang akan dilakukan
' Output:
'        Secant adalah hasil hitungan akar dengan ketelitian Eps
'        ierr adalah kode error
'        ierr = 2, fatal error, penyebut = 0
'        ierr = 1, warning, akar belum memenuhi ketelitian,
'                  namun maksimum iterasi sudah tercapai
'        ierr = 0, sukses mencari akar

Dim NoIter As Integer, Penyebut As Double, X2 As Double

' Sediakan variabel untuk counter
   NoIter = 1
   

Langkah3:
' Hitung penyebut pada metoda Newton
   Penyebut = (f(X1) - f(X0)) / (X1 - X0)

' Jika penyebut = 0, maka error terjadi
   If Penyebut = 0# Then
       ierr = 2
       Exit Function
   End If
   

' Mulai metoda Secant
   X2 = X1 - f(X1) / Penyebut
   Secant = X2

' Test konvergensi
   If Abs(X2 - X1) <= eps Then
       ierr = 0
       Exit Function
   End If

' Jika iterasi terlampaui, maka selesailah tugas kita
   If NoIter = itmax Then
       ierr = 1
       Exit Function
   End If
  

' Jika belum memenuhi kriteria lakukan hitungan lagi
   NoIter = NoIter + 1
   X0 = X1
   X1 = X2
   GoTo Langkah3
   

End Function


